
ABSTRACT
This is a proposal for a bit-precise word-level format, called
BTOR. It is easy to parse and has precise semantics. In
its basic form it allows to model SMT problems over the
quantifier-free theory of bit-vectors in combination with onedimensional arrays. Our main contribution is a sequential
extension that can be used to capture model checking problems on the word-level. We present two case studies where
BTOR is used as sequential format. Finally, we report on
experimental results for the model checking extension of our
SMT solver Boolector.

Categories and Subject Descriptors
B.5.2 [Register-Transfer-Level Implementation]: Verification; D.2.4 [Software Engineering]: Software/Program
Verification; F.4.3 [Mathematical Logic and Formal Languages]: Decision Problems

1.

INTRODUCTION

To improve the state-of-the-art in automatic reasoning,
standardized benchmarks are required. As a first step in
producing such benchmarks one has to agree on an input
format or at least on semantics. Not only for the developers,
but also for the users of automatic reasoning tools, a simple
standard format with precise semantics is of great value.
Prominent examples of such formats are, first of all, the
DIMACS format [15] for SAT solvers, the TPTP format
[22] for automatic theorem provers, the BAT format [18] for
machine descriptions and LTL specifications, the Spear Format (SF) [3] for bit-vectors, and the CVC [1] and SMT-LIB
format [20] for SMT solvers. More recently, the second author proposed a standard format for hardware model checking, called AIGER [5], which is used in the hardware model
checking competition (HWMCC).

In the domain of model checking, with the exception of
AIGER, every model checker seems to come with its own
format. Differences in syntax and particularly in semantics make it very difficult to compare model checkers. For
instance, in hardware model checking, common formats include SMV format [19], BLIF resp. BLIF-MV [8], and of
course hardware description languages (HDLs) such as Verilog and VHDL. Similar arguments apply to software model
checking formats like Promela [16], and real programming
languages like Java [2, 23] and C [4, 13].
Even just for the SMV format, multiple interpretations exist on the precise semantics of the model, and in particular
on how properties are interpreted. For example, consider
an SMV model with an at least partial relational encoding of the transition relation in a TRANS section. This is
typical for monitor constructions that model environment
constraints, e.g. the Intel benchmarks in the HWMCC.
Consider the case where we want to check a simple safety
property AGp. A counter example for this property consists
of a path from an initial state to a bad state in which p does
not hold. However, according to the original SMV semantics only those bad states are real bad states from which
it is still possible to continue execution afterwards forever
(EG1 holds in addition to ¬p). In particular, a state after
which immediately, or after a while, no further transition is
possible is not considered as a real bad state.
Various model checkers that accept the SMV input format
treat this problem differently: one tool concludes that the
property holds, the other not. It is also not always feasible to enforce the original SMV semantics, as the tool has
to be able to check EG1, which for instance bounded model
checkers can not. Various similar glitches exist among model
checkers accepting the SMV input language. These problems become even more severe as soon as models have to
be translated from one “rich” input language into the input
language of another model checker.
In the AIGER format, the solution is to restrict the semantics to the least common denominator accepted by most
model checkers. This solves issues with different interpretation of initial states and bad states. Some of the design
decisions were to always enforce every latch resp. register to
be zero initialized, outputs to be just bad state detectors,
only one operation (AND) being supported, and last but not
least, everything being represented on the bit-level.
Already the first time the AIGER format was discussed
at the Alpine Verification Meeting (AVM’06), there was a
hot debate, whether this last design decision, i.e. keeping
everything on the bit-level, should not be taken back. We

still think that a plain bit-level format has many advantages.
The semantics are simpler than those of any word-level format. The developers can concentrate on improving the core
algorithms. Furthermore, it is easier to convert benchmarks.
However, there are various techniques that can benefit
from additional information which the word-level structure
provides. This is one of the main arguments for applying
SMT techniques to bit-precise problems. Our proposal for a
new bit-precise format applies the same argument to wordlevel model checking.
Our proposed format BTOR is closer to an HDL language
than the bit-vector format of SMT-LIB, for which BTOR
can actually be seen as an alternative. However, the SMTLIB format can not capture model checking problems. Finally, other alternative input formats, such as HDLs, wordlevel input languages of model checkers, or even programming languages, all have complex semantics. Translation,
and even just parsing, is complex and error prone.
In order to foster research in model checking, it is important to have a simple and precise standardized format. A
simple format places the burden of encoding a rich format
and proper handling of complex semantics on the person
resp. tool that generates the benchmark. However, it encourages the development of model checkers and thus will
speed-up improving the state-of-the-art in model checking.

2.

OVERVIEW

BTOR is a quantifier-free word-level format for formulas over bit-vectors in combination with one-dimensional arrays. In principle, it is a world-level generalization of the
AIGER [5] format. It is strongly typed, easy to parse, multirooted, and has precise semantics. Unlike SF, which only
supports bit-vectors up to 64 bits, bit-vectors in BTOR can
have an arbitrary bit-width.
BTOR supports bit-vector variables, constants, and onedimensional bit-vector arrays. Most bit-vector operations
can be used in a signed or unsigned context. In the signed
context, bit-vectors are interpreted as being represented in
two’s complement.
Similar to SF, the result of every operation is assigned to
an intermediate variable, as the following example shows:
1 var 32
2 var 32

3 constd 32 127
4 and 32 1 -2

5 eq 1 3 4
6 root 1 5

In lines 1 and 2 we declare two 32-bit variables. In BTOR,
a non-negative integer in the first column is used as unique
identifier. Typically, we use the line number as identifier.
In line 3 we declare the decimal 32-bit constant 127. Line 4
shows how computation is expressed in BTOR. We use the
bit-wise operator and and apply it to the operands with the
identifiers 1 and 2, which are the variables we have declared
before. The result of this operation has the identifier 4. The
minus before 2 expresses logical negation 1 , i.e. we flip all
bits of the variable 2 before we combine it with the variable
1. The result has 32 bits, which is indicated by the bitwidth after the operator. In line 5 we compare the decimal
constant with the subformula 4 for equality. The length of
this result is 1, as relational operators are boolean. Finally,
we declare the subformula 5 as boolean root.
This simple example already shows some interesting design decisions. The format should be easy to parse. The first
1
Note that BTOR also supports the unary but redundant
operator not.

column is used for the identifier, the second for the operator, and the third for the bit-width. This restriction makes
it easy to write a parser for BTOR. The additional type information in the third column makes it easy for the parser
to check type consistency on the fly.
Being able to add back annotation in form of an explicit
symbol table simplifies many applications. In BTOR there
is no separate symbol table as in AIGER. However, variable
declarations can have an additional symbol part, separated
by white space after the bit width. Comments start with ‘;’
and stretch until the end of the line.
BTOR does not allow forward references, i.e. all operands
have to be declared before they are used. This restriction
makes parsing BTOR instances trivial, as it can be done in
one pass. For example, our visualizer of the BTOR format
consists of an AWK script of less than 80 lines, and produces
a graph description in DOT format. A utility that prints
a histogram of the number of operator occurrences can be
implemented as the following simple AWK script:
{a[$2]++} END {for(k in a) \
printf "%-7s%d\n", k, a[k] | "sort -n -k 2";}
As in Verilog, boolean variables are treated just as bitvectors with bit-width one. This simplifies the format as we
do not have to convert from boolean to bit-vector and vice
versa. We can simply treat the boolean case as bit-vector
case with bit-width one.
BTOR is multi-rooted, which allows to model multiple
outputs. Multiple roots can be used to model hardware
systems, which typically have multiple bit-vector outputs.

3.

BIT-VECTORS

BTOR supports the following bit-vector constructors: var
for bit-vector variables, constd, consth and const for decimal, hexadecimal and binary constants, and one, ones and
zero for the constants 1, -1 and 0. All constructors take
the bit-width as first argument. The constructor var takes
an optional symbol string as second argument. The constructors constd, consth and const take the constant value
as second argument.
The set of bit-vector operators is shown in tables 1, 2, 3
and 4. The columns w1 to w3 represent the bit-width of the
operands. The column wr represents the bit-width of the
result. The semantics of most operators are defined by the
semantics of the corresponding operators in the quantifierfree theory of fix-sized bit-vectors QF_BV in the SMT-LIB
standard [20]. The only exceptions are as follows.
The SMT-LIB standard does not specify the result of dividing by zero. In BTOR the result of dividing by zero is
the largest unsigned integer that can be represented with the
bit-width of the operands. This corresponds to real divider
circuits in hardware systems. Nevertheless, the underspecified variant of the SMT-LIB format can always be modelled
by treating divison by zero as uninterpreted function.
The bit-width of shift operands are restricted in the following way. The bit-width of the first operand has to be a
power of two. The bit-width of the second argument has to
be log 2 of the bit-width of the first operand. If the bit-width
of the second shift operand is log 2 , then it is impossible to
shift more than the bit-width, which for example is undefined in the programming language C [10, 17].
Additionally, BTOR supports the Verilog reduction operators redand, redor and redxor, the VHDL rotate operators

class
negation
reduction
arithmetic

operators
not, neg
redand, redor, redxor
inc, dec

w1
n
n
n

wr
n
1
n

class
conditional

Table 1: The unary bit-vector operators not and
neg apply one’s resp. two’s complement. The operators redand, redor and redxor are reduction operators from Verilog. Finally, inc and dec are used to
increment resp. decrement a bit-vector by one.
class
bitwise
boolean
arithmetic
relational
shift
overflow
concat

operators
and, or, xor,
nand, nor, xnor
implies, iff
add, sub, mul, urem
srem, udiv, sdiv, smod
eq, ne, ult, slt,
[us]lte, [us]gt, [us]gte
sll, srl, sra, ror, rol
[us]addo, [us]subo,
[us]mulo, sdivo
concat

w1 w2

wr

n

n

n

1

1

1

n

n

n

n

n

1

n log2 n

n

n

n

1

n1

n2 n1 + n2

Table 2: Binary bit-vector operators. Some operators can be used in a signed or unsigned context,
e.g. sdiv represents signed bit-vector division with
two’s complement semantics. For every arithmetic
operator there is a corresponding overflow detection
operator. The only exception is udiv as unsigned division can never overflow. Signed division overflows
if we divide the smallest negative integer by -1.

1 var 32
2 var 32
3 redand 1 1

4 redand 1 2
5 umulo 1 1 2
6 and 1 3 4

7 and 1 6 -5
8 root 1 7

The reduction operator redand returns 1, if all bits of the
operand are set to one. In this case unsigned multiplication
overflows. We assert the opposite in line 7, which makes this
instance unsatisfiable.
Whether we want to detect arithmetic overflows or not,
depends on the application scenario. Typically, in software
verification we are interested in detecting overflows, as the
results are often undefined and depend on compliler semantics [10]. However, in hardware verification overflow detection is in most cases unnecessary. In our C expression
checker C32SAT [10] overflow detection is always automatically applied, which makes verification instances unnecessarily hard if we do not care about overflows. Thus, we separated overflow detection from arithmetic, and introduced
an additional set of overflow detection operators in BTOR.
We get only harder verification instances if we are interested
in overflow detection.
Finally, it is of course possible to express some of our
operators in terms of others. For instance our SMT solver
Boolector [11, 12] supports the full set of operators, both

w1
1

w2
n

w3
n

wr
n

Table 3: The only ternary bit-vector operator cond
represents a functional if-then-else. If the condition
is 1, it returns the second argument, and the third
argument otherwise.
class
extract

operators
slice

w1
n

upper
u

lower
l

wr
u-l+1

Table 4: Miscellaneous bit-vector operators. The
first operand identifies the variable to which slice is
applied, upper and lower are immediates. The slice
operator is the only operator that uses immediates.
in the application interface (API) and of course also in the
parser for the BTOR format. Internally, however, we only
use a sub-set of base operators, which are shown underlined
and bold. Our selection can be considered to be arbitrary. It
is motivated by the kind of word-level rewriting implemented
in Boolector, which could be different for other solvers.
We can imagine various word-level techniques that can
benefit from non-base operators. However, in contrast to
the bit-level format AIGER, it is much harder to extract
non-base word-level operators after they have been represented with base operators. Therefore, we argue that in a
general format for bit-precise word-level modelling it should
be possible to express and retain non-base operators. Nevertheless, it is always possible to rewrite benchmarks that
contain non-base operators into the base format.

4.
rol and ror, and a new set of overflow detection operators.
For example, the overflow detection operator umulo returns
1 if unsigned multiplication overflows. Consider the following example:

operators
cond

ARRAYS

Arrays can be constructed with the operator array. The
first argument is the bit-width of the elements, and the second the bit-width of the addresses. BTOR supports bitvector arrays in combination with the array operations read,
write, acond and eq. Equality can be applied to arrays and
array elements, i.e BTOR supports the extensional theory
of arrays. Consider the following example:
1
2
3
4
5
6
7

array 32
array 32
array 32
var 4
var 32
var 1
acond 32

4
4
4

8 write 32 4 7 4 5
9 read 32 8 4
10 eq 1 5 9
11 eq 1 3 8
12 and 1 10 11
13 root 1 12

4 6 1 2

In the first three lines we declare three arrays with element
bit-width 32, and size 24 = 16. In line 7 we use an if-thenelse on the arrays 1 and 2. If condition 6 holds, we return
1, and 2 otherwise. The result is an array with bit-width
32 and size 16. Line 8 writes on 7 the element 5 at index
4. The result is an array, where the element at position 4
is overwritten with 5. The elements on all other positions
remain the same. In line 9 we read 32 bits on 8 at index
4, i.e. we read the value that has been written at this index
before. In line 10 we compare the read value with 5. In line
11 we compare the arrays 3 and 8. Finally, we assert 10 and
11, and declare the result as root.

5.

SEQUENTIAL EXTENSION

BTOR supports modelling sequential and synchronous circuits with registers and memories. Registers are modelled
with the help of var and next, and memories with array and
anext. Variables without next, and arrays without anext, are
primary inputs, which are fresh in every clock cycle.
Registers are initialized to zero. Memories are uninitialized as this is the general case for main memory in software,
and memory units in hardware. Consider the following:
1 var 32
2 var 32

3 xor 32 1 2
4 next 32 1 4

In line 5 we apply next to 1, which determines that 1 is a
register. Variable 2 is an input as there is no next function
applied to it. The next function applied to 1 is 4, i.e. in every
cycle we apply xor to the current content of the register 1,
and the primary input 2.
Additionally, BTOR supports modelling safety properties,
which can be used for model checking. A safety property is
represented by a root with bit-width one. Multiple roots
with bit-width one are implicitly disjuncted as they represent different bad states. Another example with registers,
one memory, and a safety property is shown in Fig. 1.

6.

CASE STUDIES

We present two case studies where we show how BTOR,
in particular its sequential extension, can be used to model
hardware and software systems. In the first case study a
FIFO with internal memory is modelled in BTOR. In the
second case study we show how a program in a BASIC-like
programming language can be linearly modelled in BTOR.

6.1

FIFOs

We took two different Verilog implementations of a typical hardware FIFO as shown in Fig. 2, and modelled both
with BTOR. In this case we manually translated the Verilog
models to BTOR, but in principle this could be automated.
The first implementation organizes its internal memory as
stack, while the second implementation uses a queue. Both
implementations use head resp. tail registers which hold the
address of the first resp. last element. The following Verilog
code fragment shows how the queue-based implementation
behaves if dequeue is active:
if (empty == 1’b0) begin
data_out <= # 1 mem[head];
head <= # 1 head + 3’b001;
end
if (head + 3’b001 == tail) begin
empty <= # 1 1’b1;
end
full <= # 1 1’b0;

Consider the following BTOR fragment for a queue-based
FIFO implementation. It shows how the next-operator can
be used to model internal memory.

1
2
3
4
5

array 8 32
var 32 start
var 32 end
var 32 p
var 1 flag

7 var 1 full
8 var 1 empty
9 var 32 data_out
10 var 3 head
11 var 3 tail
...
...
56 acond 32 3 5 55 6
57 acond 32 3 1 56 6
58 anext 32 3 6 57

; next(memory)
22 write 8 32 1 4 6
23 eq 1 5 8
24 ult 1 4 3
25 and 1 23 24
26 acond 8 32 25 22 1
27 anext 8 32 1 26

6 zero 8
7 one 32
8 one 1
; next(start)
9 var 32 start_in
10 ne 1 5 8
11 cond 32 10 9 2
12 next 32 2 11

; check property
28 var 32 i
29 read 8 1 28
30 ne 1 29 6
31 ult 1 2 3
32 ult 1 28 2
33 ult 1 28 3
34 and 1 -32 33
35 and 1 34 -10
36 and 1 31 35
37 and 1 30 36
38 eq 1 4 3
39 and 1 38 37
40 root 1 39

; next(end)
13 var 32 end_in
14 cond 32 10 13 3
15 next 32 3 14
; next(p)
16 add 32 4 7
17 ult 1 4 3
18 cond 32 17 16 4
19 cond 32 10 9 18
20 next 32 4 19

Figure 1: BTOR encoding of function memclear
which sets a memory region from location start up to
but not including end to zero. Memory is modelled
as an 8-bit bit-vector array containing 232 elements
(line 1). In the first clock cycle indicated by register
flag (lines 5 and 21), registers start and end are set
to values start in and end in (lines 9-12 and 13-15),
which represent user inputs, and remain constant
afterwards. Value zero is written to memory (lines
22-27) at location p, which is first set to start and
then incremented (lines 16-20) until end. Bad states
(lines 28-40) represent that it is possible to read a
non-zero value from a location between start and
end after clearing memory has finished. The resulting instance is unsatisfiable.

reset
enqueue

1 var 1 reset
2 var 1 enqeue
3 var 1 deqeue
4 var 32 data_in
5 xor 1 2 3
6 array 32 3
...
53 write 32 3 6 11 4
54 acond 32 3 7 6 53
55 acond 32 3 2 54 6

; next(flag)
21 next 1 5 8

dequeue
data_in

empty

FIFO

data_out
full

Figure 2: Hardware FIFO as black box. The clock
signal is omitted.

Instruction
ADD arg
NE arg
JMP arg
EXIT arg
SAVE arg

Semantics
accu := accu + arg
flag := (accu 6= arg)
if flag then pc := arg
exit with code arg
arg := accu

Table 5: Subset of the instruction set of a BASIClike assembler for a hypothetical accumulator architecture. Instructions listed in the table are used in
the example in the text. All instructions are unary.
Address modes are register or immediate. There are
26 unsigned integer registers named a to z. A special register flag is used for conditional branches,
which is set by a preceding relational instruction.
Instructions LOAD, SAVE, PEEK and POKE read and write
registers and memory locations, respectively. EXIT
terminates the program with specified exit code.

In line 58 the next-operator is applied to the array, which
models internal FIFO memory. The if-then-else chain can
be read as follows. If reset is active (low), or enqueue and
dequeue are equal, or enqueue is not active, or enqueue is
active and the FIFO is full, memory remains in the same
state. The if-then-else chain models the priority of the signals. However, if enqueue is active and the FIFO is not full,
we write at the position the register tail points to the current input data in. Analogously, we modelled the bit-vector
registers head, tail, empty, full and data out.

6.2

Checking assembly-level programs

As demonstrated in Fig. 1, model checking problems can
be described in our BTOR format. We propose to model
assembly-level programs in BTOR, where memory and registers are represented as bit-vector arrays. Program states
can be related to each other by the next-operator on nested
conditional expressions involving the program counter. Assertions can be regarded as safety properties. Violations can
be detected by checking if bad states are reachable.
For demonstration purposes, we implemented a simulator and translator for a very simple, BASIC-like assembler
for a hypothetical accumulator architecture as described in
Tab. 5. An assembler program is translated into a BTOR
model. The size of the model is linear in the size of the
program. Failed assertions terminate the program with a
non-zero exit code.
Consider the following trivial assembler program:
10 load 100
20 add 1
30 save a

40 ne a
50 jmp 70
60 exit 0

70 exit 1

After the accumulator has been assigned value 100 and incremented by one, its value is stored in register a (lines 10-30).
If the accumulator value does not equal the value stored in
a, then the program terminates with a non-zero exit code
(lines 40, 50, 70). This behavior is regarded as an assertion
failure. Fig. 3 shows the corresponding BTOR model.
In principle, our approach can be extended to subsets
of more sophisticated assembly languages and architectures
like x86.

1 var 32 pc
2 var 32 accu
3 var 1 flag
4 array 32 5
5 array 32 32
6 constd 32 0
7 constd 32 10
8 constd 32 20
9 constd 32 30
10 constd 32 40
11 constd 32 50
12 constd 32 60
13 constd 32 70
14 constd 5 0
15 write 32 4 14 2
16 read 32 4 14
17 eq 1 1 6
18 cond 32 17 7 1
19 constd 32 10
20 eq 1 1 7
21 cond 32 20 8 18
22 constd 32 100
23 cond 32 20 22 2
24 eq 1 1 8
25 cond 32 24 9 21
26 constd 32 1
27 add 32 23 26
28 cond 32 24 27 23

29
30
31
32
33
34
35
36
37
38
39
40
41
42
43
44
45
46
47
48
49
50
51
52
53
54
55
56

eq 1 1 9
cond 32 29 10 25
write 32 5 4 14 2
acond 32 5 29 31 4
eq 1 1 10
cond 32 33 11 30
ne 1 2 16
cond 1 33 35 3
eq 1 1 11
cond 32 37 12 34
and 1 37 3
cond 32 39 13 38
constd 32 70
eq 1 1 12
constd 32 0
redor 1 43
and 1 44 42
root 1 45
eq 1 1 13
constd 32 1
redor 1 48
and 1 49 47
root 1 50
next 32 1 40
next 32 2 28
next 1 3 36
anext 32 5 4 32
anext 32 32 5 5

Figure 3: BTOR model of the simple assembler program. The program counter pc can assume values
for each line in the source code (lines 6-13), with
zero as initial value. Its next-values are determined
in nested conditional expressions (lines 52, 40, 38,
34, 30, 25, 21 and 18). If register flag is set at the JMP
instruction (lines 37-40), then the new pc will equal
70, i.e. the jump to EXIT 1 is executed. This kind of
program termination corresponds to the failed assertion REGS[a] = 101 in line 40 of the program. If
pc = 30, then the accumulator value is written to
register a, which has index 0 (line 14) in the bitvector array modelling the register set (lines 55, 32,
31, 29). All other registers remain constant. In this
example, memory is not written and remains constant (line 56), but the idea for modelling memory
writes is similar as for registers. The accumulator is
assigned value 100 and incremented by one (lines 53,
28, 23). Exit code zero designates normal program
termination.

k
00
01
02
03
04
05
06
07
08
09
10

boolector
inc
non-inc
adc noadc
adc noadc
0.0
0.0
0.0
0.0
0.1
0.0
0.1
0.0
0.8
0.1
0.9
0.1
15.4
0.3
6.1
0.4
41.9
0.7
38.0
1.1
91.2
1.6
96.5
2.7
419.8
3.7
267.8
5.8
to
6.8
to
11.7
to
14.3
to
23.9
to
31.1
to
47.8
to
73.7
to
97.2

z3
non-inc
adc noadc
0.0
0.0
0.0
0.0
300.3
0.0
639.9 313.1
to
to
to
to
to
to
to

to
to
to
to
to
to
to

Table 6: Experiments for equivalence checking of the
FIFOs in 6.1. We compare our SMT solver Boolector with Z3 1.2. We ran our benchmarks on our
cluster of 3 GHz Pentium IV with 2 GB main memory, running Ubuntu Linux. Time limit is 900 seconds and memory limit is 1500 MB. Time out is
indicated by to.

7.

EXPERIMENTS

We extended our SMT solver Boolector [11, 12] for checking sequential BTOR, i.e. we implemented a bounded model
checker within Boolector. Boolector uses a functional AIG
encoding including two level AIG rewriting [9]. PicoSAT [6]
is used as SAT solver. Boolector supports bounded model
checking for witnesses [7], and k-induction [21] with and
without All Different Constraints (ADCs). ADCs are used
to represent simple path constraints. Note that model checking instances with memories generates ADCs with inequalities on arrays, which need extensionality [11, 12].
We compared Boolector’s incremental model checking with
non-incremental Boolector and Z3 1.2 [14]. The benchmarks
are parametrized instances for checking behavioral equivalence of two FIFO implementations as presented in 6.1. Bitwidth and size of memory is 32. The results are shown in
table 6. Column 1 shows the upper bound of the model
checking instance. Column 2 shows the time of incremental
model checking, which includes searching for witnesses and
k-induction with ADCs. In column 3 ADCs are disabled.
The non-incremental results have been computed in the
following way. First, we generated SMT instances that represent (i) search for witnesses, (ii) k-induction with ADCs,
and (iii) k-induction without ADCs. In columns 4 and 6 we
summed up the solving times for the results of (i) and (ii),
from 0 to k. Analogously, we computed column 5 and 7 by
summing up the results of (i) and (iii). Both versions of
Boolector clearly outperform Z3. In the case where ADCs
are disabled, the incremental model checking of Boolector is
faster than the non-incremental.

8.

CONCLUSION

We proposed a bit-precise word-level format BTOR. It is
easy to parse, has precise semantics, and allows to model
SMT problems over the theory of bit-vectors in combination with one-dimensional arrays. Our main contribution is
a sequential extension of BTOR to model registers and memories. We discussed two sequential case studies. Finally, our

experiments show that our incremental model checker can
efficiently decide model checking problems in BTOR.

9.

